/*-----------------------------------------------------------------------------
	The contents of this file are subject to the Mozilla Public License
	Version 1.1 (the "License"); you may not use this file except in
	compliance with the License. You may obtain a copy of the License at
	http://www.mozilla.org/MPL/
	
	Software distributed under the License is distributed on an "AS IS"
	basis, WITHOUT WARRANTY OF ANY KIND, either express or implied. See the
	License for the specific language governing rights and limitations
	under the License.
	
	The Original Code is Fever Framework code.
	
	The Initial Developer of the Original Code is Romain Ecarnot.
	Portions created by Initial Developer are Copyright (C) 2006
	the Initial Developer. All Rights Reserved.
	
	Contributor(s):
	
	Unless required by applicable law or agreed to in writing, software
	distributed under the License is distributed on an "AS IS" BASIS,
	WITHOUT WARRANTIES OR CONDITIONS OF ANY KIND, either express or implied.
	See the License for the specific language governing permissions and
	limitations under the License.
-----------------------------------------------------------------------------*/

import com.bourre.data.collections.Queue;

import fever.structures.QueueProtectionType;
import fever.utils.ArrayUtil;

/**
 * Queue structure with protected {@link #enqueue()} method.
 * 
 * <p>2 protections are available : 
 * <ul>
 *   <li>Instance protection ( {@link fever.structures.QueueProtectionType#INSTANCE} )</li>
 *   <li>class type protection ( {@link fever.structures.QueueProtectionType#TYPE} )</li>
 * </ul>
 *
 * @author Romain Ecarnot
 */
class fever.structures.ProtectedQueue extends Queue
{
	//-------------------------------------------------------------------------
	// Private properties
	//-------------------------------------------------------------------------
	
	private var _protectionFunc : Function; 
	
	 	
	//-------------------------------------------------------------------------
	// Public API
	//-------------------------------------------------------------------------
	
	/**
	 * Constructor.
	 * 
	 * @param protection Queue protection type 
	 * (default {@link fever.structures.QueueProtectionType#INSTANCE}
	 */
	public function ProtectedQueue( protection : QueueProtectionType )
	{
		super();
		
		_protectionFunc = ( protection == QueueProtectionType.TYPE )
			? _hasType
			: _hasInstance;
		
	}
	
	/**
	 * Adds passed-in {@code o} element to queue if not already in.
	 * 
	 * <p>Checks constructor to know how protection works.
	 * 
	 * @param o Element to add
	 */
	public function enqueue( o ) : Void
	{
		if( !_protectionFunc( o ) )	super.enqueue( o );
	}
	
	/**
	 * Removes passed-in {@code o} element from queue.
	 * 
	 * <p>Be careful ! it's not the normal queue use.<br />
	 * Usually use the {@code dequeue} method to remove element from queue.<br />
	 * {@link #remove()} allow queue preparation before any normal queue process.
	 */
	public function remove( o ) : Void
	{
		ArrayUtil.remove( o, _a );
	}
	
	//-------------------------------------------------------------------------
	// Private implementation
	//-------------------------------------------------------------------------
	
	private function _hasInstance( o ) : Boolean
	{
		return ArrayUtil.contains( o, _a );
	}
	
	private function _hasType( o ) : Boolean
	{
		return ArrayUtil.containsType( o, _a );	
	}	
}